

PRÁCTICA 3. CURSO 2016-2017.

INTEGRALES DE LÍNEA Y

DE SUPERFICIE.

1 Integral de línea

Vamos a resolver en esta práctica las cuestiones habituales que nos aparecen cuando tratamos de resolver una integral de línea. Lo normal para estos problemas es que nos pidan:

- Representar la curva (en 2D o 3D).
- Comprobar si la integral de línea es independiente del camino de integración.
- Si depende del camino, calcular la integral tomando una parametrización del camino; si el camino es cerrado, calcular la integral usando el th de Green (si es una curva en 2D) o el th de Stokes (si es curva en 3D).
- Si no depende del camino, y el camino es cerrado, sabemos que da 0; si el camino no es cerrado, podemos resolverla cambiando el camino por otro más simple (tomando una parametrización de éste último), o usando la función potencial.


Vamos a realizar todas las cuestiones anteriores con ejemplos:

EJEMPLO 1: Dado el campo vectorial

$F(x,y) = (e^x \sin(y) - y, e^x \cos(y) - x - 2)$ y la curva dada por $\gamma(t) = (1+t/2\pi) (\cos(t), \sin(t))$, con t en $[0, 2\pi]$, se trata de:

1.a Representar γ :

```
--> wxplot2d([parametric, (1+t/2*pi)*cos(t), (1+t/2*pi)*sin(t),
[t,0,2*pi],[nticks,50]]);
```



1.b Probar que la integral de F a lo largo de γ es independiente del camino:

En este caso hay que ver que si denotamos por $P(x,y) = e^x \sin(y) - y$, $Q(x,y) = e^x \cos(y) - x - 2$, se tiene que $d/dx(Q) = d/dy(P)$, o lo que es lo mismo, que $d/dx(Q) - d/dy(P) = 0$:

```

--> /*Calculamos d/dx(Q)-d/dy(P)*/
diff(exp(x)*cos(y)-x-2,x,1)-diff(exp(x)*sin(y)-y,y,1);
(%o2) 0

```

1.c Calcular directamente la integral de línea:
En primer lugar definimos el campo F:

```

--> F(x,y):=[exp(x)*sin(y)-y,exp(x)*cos(y)-x-2];
(%o3) F(x,y):=[exp(x)sin(y)-y,exp(x)cos(y)-x-2]

```

Como tenemos que sustituir "x" por la 1ª componente de gamma e "y" por la segunda, hacemos:

```

--> x:(1+t/2*%pi)*cos(t);
y:(1+t/2*%pi)*sin(t);
(%o4)  $\left(\frac{\pi t}{2}+1\right)\cos(t)$ 
(%o5)  $\left(\frac{\pi t}{2}+1\right)\sin(t)$ 

```

Ahora tenemos que multiplicar escalarmente F por el vector [dx/dt,dy/dt], y a su resultado calcularle la integral en [0,2pi]:

```

--> integrate(F(x,y).[diff(x,t,1),diff(y,t,1)], t, 0, 2*%pi);
(%o6) 0

```

1.d Calcular esta integral sustituyendo gamma por un camino más sencillo:
En nuestro caso vamos a tomar la recta que une los puntos origen (1,0) y extremo (2,0); es decir, tomamos un nuevo camino que tiene por parametrización (t,0), con t en [1,2] y realizamos lo anterior:

```

--> x:t$
y:0$
integrate(F(x,y).[diff(x,t,1),diff(y,t,1)], t, 1, 2);
(%o9) 0

```

1.e Comprobar este resultado usando la función potencial:
En este caso se trata de calcular una función fi(x,y) tal que d/dx (fi)=P(x,y) y d/dy (fi)=Q(x,y); y entonces sabemos que el valor de la integral será fi(2,0)-fi(1,0)

```

--> /*en primer lugar calculamos la función potencial fi integrando
P(x,y); así la obtendremos salvo una función g(y); OJO:
primero tenemos que borrar el valor dado a las variables x e y
ya que anteriormente les hemos dado los valores, respect. t y o*/

remvalue(all)$

/*Esto suele hacerse para eliminar posibles
valores que le hallamos podido dar a las incógnitas que hemos
usado hasta ahora*/

integrate(exp(x)*sin(y)-y, x);
(%o11) %ex sin(y)-x y

```

De momento ya tenemos que $f_i(x,y) = e^x \sin(y) - x*y + g(y)$, y tenemos que determinar esta $g(y)$. Para ello, derivamos este resultado respecto de y , $d/dy(f_i)$

```
--> diff(%,y,1);
(%o12) %e^x cos(y)-x
```

y si comparamos lo obtenido con $Q(x,y)$, resulta que $g'(y)=-2$, por lo que $g(y)$ será

```
--> integrate(-2*y, y);
(%o13) -y^2
```

Por tanto, se tiene que $f_i(x,y)=e^x \sin(y)-x*y - y^2$, y el resultado final será el mismo obtenido anteriormente:

```
--> fi(x,y):=%e^x*sin(y)-x*y - y^2;
      fi(2,0)-fi(1,0);
(%o14) fi(x,y):=%e^x sin(y)-x y-y^2
(%o15) 0
```

EJEMPLO 2: Calcular la integral del campo vectorial $G(x,y)=((1+x)y^2, \pi x - \log(1+y^2) \cos(y^4+3y))$ a lo largo de la curva cerrada, orientada positivamente, dada por $(x-1)^2+(y-1)^2=4$:

```
--> /*en primer lugar vemos si es independiente del camino*/

      remvalue(all);
      P(x,y):=(1+x)*y^2;
      Q(x,y):= pi*x - log(1+y^2)*cos(y^4+3*y);
(%o16) []
(%o17) P(x,y):=(1+x)y^2
(%o18) Q(x,y):=pi x-log(1+y^2)cos(y^4+3 y)
```

```
--> diff(Q(x,y),x,1)-diff(P(x,y),y,1);
(%o19) pi-2(x+1)y
```

y como la diferencia anterior es no nula, resulta que la integral es dependiente del camino. Como éste es cerrado, aplicaremos el th de Green en el plano. Así sabemos que la integral que queremos calcular coincide con la integral doble de la diferencia anterior, por lo que tendremos que calcular la integral doble de $\pi-2*(x+1)*y$ en el círculo centrado en el punto $(1,1)$ y de radio 2. Evidentemente nos va a interesar realizar un cambio a polares, por lo que si hacemos $x=1+r*\cos(t), y=1+r*\sin(t)$, tendremos:

```
--> f(x,y):=%pi-2*(x+1)*y;
(%o20) f(x,y):=pi-2(x+1)y
```

```

--> x:1+r*cos(t);
      y:1+r*sin(t);
(%o21) r cos(t)+1
(%o22) r sin(t)+1

--> integrate(integrate(f(x,y)*r,r,0,2),t,0,2*%pi);
(%o23)  $-\frac{48\pi-12\pi^2}{3}$ 

```

□

2 Integral de superficie

Veamos ahora como wxMaxima nos ayuda a resolver integrales de superficie. También lo haremos con ejemplos:

EJEMPLO 3: Dado el campo vectorial $G(x,y,z)=(-y,y*z^2,x^2*z)$ calcular el flujo de su rotacional a través del hemisferio superior de la esfera centrada en el origen y de radio 4:

Comenzamos obteniendo el rotacional de G:

```

(%i1) remvalue(all);
(%o1) []

(%i2) load(vect)$
      G: [-y,y*z^2,x^2*z];
      express(curl(G));
      ev(%, diff);
(%o3) [-y,y z^2,x^2 z]
(%o4) [ $\frac{d}{dy}(x^2 z) - \frac{d}{dz}(y z^2)$ ,  $\frac{d}{dz}(-y) - \frac{d}{dx}(x^2 z)$ ,  $\frac{d}{dx}(y z^2) - \frac{d}{dy}(-y)$ ]
(%o5) [-2 y z, -2 x z, 1]

```

y para que nos de su expresión en forma de vector (que llamaremos rotG):

```

(%i6) rotG:%;
(%o6) [-2 y z, -2 x z, 1]

```

Maxima no calcula directamente integrales de superficie (al igual que no calcula integrales de línea, como hemos visto), sino que tenemos que pasar la integral de superficie a una integral doble (de forma similar a como hemos transformado una integral de línea en una integral simple) y entonces calcularemos ésta última. Para ello, introducimos la ecuación de la superficie y calculamos su vector normal:

```
(%i7) S(x,y,z) := x^2+y^2+z^2-4;
      express(grad(S(x,y,z)));
      gradS:ev(%, diff);
(%o7) S(x,y,z) := x^2+y^2+z^2-4
(%o8) [ $\frac{d}{dx}(z^2+y^2+x^2-4)$ ,  $\frac{d}{dy}(z^2+y^2+x^2-4)$ ,  $\frac{d}{dz}(z^2+y^2+x^2-4)$ ]
(%o9) [2 x, 2 y, 2 z]
```

Este vector tenemos que dividirlo por su módulo, para después multiplicarlo escalarmente por rotG.

```
(%i10) rotG.gradS;
(%o10) 2 z-8 x y z
```

No hemos dividido por el módulo porque conocemos que este denominador desaparecerá al multiplicar con posterioridad por la expresión de dS (que si hacemos aparte las operaciones, obtenemos la misma raíz del módulo pero dividido por $2z$ -estamos usando para dS la expresión en la que la ecuación de S viene dada en forma implícita y donde se proyecta sobre el plano OXY -); pero sí que hemos de dividir la expresión anterior por $2z$. En resumen, al final tenemos que resolver la integral doble de $1-4xy$ en el círculo de radio 2, lo que haremos en polares (como en el último ejercicio de la sección anterior):

```
(%i11) f(x,y) := 1-4*x*y;
      x:r*cos(t);
      y:r*sin(t);
      integrate(integrate(f(x,y)*r,r,0,2),t,0,2*%pi);
(%o11) f(x,y) := 1-4 x y
(%o12) r cos(t)
(%o13) r sin(t)
Principal Value
atanh: argument 1 isn't in the domain of atanh.
-- an error. To debug this try: debugmode(true);
```

Si lo hacemos directamente vemos que nos da un mensaje de error, por lo que lo haremos por separado (es decir, primero calculamos la integral con respecto a r en el intervalo $[0,2]$ y después calcularemos la integral del resultado respecto de t):

```
(%i15) integrate((1-4*r*cos(t))*r*sin(t)*r,r,0,2);
(%o15)  $\frac{1}{16 \cos(t) \sin(t)} - \frac{256 \cos(t)^2 \sin(t)^2 - 32 \cos(t) \sin(t) + 1}{16 \cos(t) \sin(t)}$ 
```

y lo simplificamos:

```
(%i16) ratsimp(%);
(%o16) 2-16 cos(t) sin(t)
```

y entonces podemos calcular la integral respecto de t :

```
(%i17) integrate(%,t,0,2*%pi);
(%o17) 4  $\pi$ 
```

3 Instrucciones para la entrega de las prácticas

- Las prácticas hay que hacerlas en fichero Maxima (extensión .wxm) y remitirlas por email.
- Aconsejable realizar un fichero para cada práctica (llamandolo, por ejemplo, Practica 1-nombre alumno.wxm)
- En los ficheros ir explicando lo que se va a realizar, incluyendo los comentarios que sean necesarios, como en el ejemplo de la sección anterior, y que siempre van en la forma:
/*COMENTARIO QUE SE QUIERE HACER*/
- Comprobar que no hay erratas y las sentencias escritas son correctas.

4 Un ejemplo resuelto


Lo que viene a continuación es una muestra de lo que hay que remitir en el fichero de prácticas. Copiar en primer lugar el enunciado y poner la resolución del ejercicio a continuación (observar como se van introduciendo los comentarios que aclaran lo que estamos realizando en la resolución).

PROBLEMA N°. Resolver, con ayuda de wxMaxima, el Problema 1 del examen de Matemáticas II de la convocatoria de FEBRERO 2014, realizando la representación gráfica de la curva, estudiando la independencia del camino de integración y calculando la correspondiente integral.

```
--> /* PROBLEMA 1(FEBRERO 2014)*/
/* Se trata de calcular la integral del línea del campo
vectorial  $F(x,y):=(x^2+y^2, 3*x*y+\log(y^2+1))$  a lo largo
de la frontera del recinto  $4*x^2+(y-1)^2 \leq 1$ */
```

```
(%i18) /*En primer lugar representamos el recinto*/
```

```
remvalue(all)$
load(draw)$
wxdraw2d(implicit(4*x^2+(y-1)^2=1,x,-1/2,1/2,y,0,2));
```



```
(%t20)
```

```
(%o20)
```

```
(%i21) /* Definimos la función*/
```

```
F(x,y):=[x^2+y^2,3*x*y+log(y^2+1)];
```

```
(%o21) F(x,y):=[x^2+y^2,3*x*y+log(y^2+1)]
```

```
(%i22) /*Vemos si F(x,y) es independiente del camino de integración*/
/*En este caso hay que ver que d/dx(Q)=d/dy(P)*/
```

```
P(x,y):=x^2+y^2;
```

```
Q(x,y):=3*x*y+log(y^2+1);
```

```
/*calculamos d/dx(Q)-d/dy(P)*/
```

```
diff(Q(x,y),x,1)-diff(P(x,y),y,1);
```

```
(%o22) P(x,y):=x^2+y^2
```

```
(%o23) Q(x,y):=3*x*y+log(y^2+1)
```

```
(%o24) y
```

```
(%i25) /* Se cumple que es dependiente del camino, ya que la diferencia
anterior es distinta de cero. Al ser el camino cerrado, podemos
resolver la integral de línea mediante el teorema de Green,
es decir, hemos de calcular la integral doble de "y" (que es la
diferencia entre d/dx(Q) y d/dy(P)) en el recinto considerado*/
/*Como el recinto de integración es una elipse de semiejes 1/2 y 1,
centrada en el punto (0,1), haremos un cambio a coordenadas
polares generalizadas x=1/2*r*cos(t); y=1+r*sin(t), siendo el
jacobiano J= r/2 (lo calcularemos), donde r varía entre 0 y 1
y t varía entre 0 y 2*pi*/
```

```
f(x,y):=y;
```

```
x:1/2*r*cos(t);
```

```
y:1+r*sin(t);
```

```
J:determinant(jacobian([x,y],[r,t]))$
```

```
trigsimp(%);
```

```
(%o25) f(x,y):=y
```

```
(%o26) 
$$\frac{r \cos(t)}{2}$$

```

```
(%o27) r sin(t)+1
```

```
(%o29) 
$$\frac{r}{2}$$

```

```
--> /*La nueva función a integrar, en polares, será*/
```

```
f(x,y)*J$
```

```
trigsimp(%);
```

```
(%o31) 
$$\frac{r^2 \sin(t)+r}{2}$$

```

```
--> /*Por tanto, la integral a calcular será:*/
```

```
integrate(integrate(%,r,0,1),t,0,2*pi);
```

```
(%o32) 
$$\frac{\pi}{2}$$

```

5 Ejercicios a entregar

PROBLEMA 1.

Resolver, con ayuda de wxMaxima, el Problema 1 del examen de Matemáticas II de la convocatoria de SEPTIEMBRE 2016, realizando la representación gráfica del recinto, estudiando la independencia del camino de integración y calculando la correspondiente integral.

PROBLEMA 2.

Resolver, con ayuda de wxMaxima, el apartado (b) del Problema 2 del examen de Matemáticas II de la convocatoria de FEBRERO 2016, realizando las correspondientes gráficas y justificando lo realizado.